Newsgroup: it.comp.appl.access
Titolo: Risoluzione grafica

Ho necessità di aprire maschere diverse in funzione della
risoluzione grafica attiva. Da 
www.allapi.net
 ho trovato questa parte di
codice
che adattato ad Access mi aiuterebbe a risolvere il problema:
ResWidth = Screen.Width \ Screen.TwipsPerPixelX
ResHeight = Screen.Height \ Screen.TwipsPerPixelY
ScreenRes = ResWidth & "x" & ResHeight
Inserito in una maschera di prova di VB5 funziona correttamente, ma
quando la inserisco in un modulo di Access (2000) non va.
Sicuramente occorrerà correggere il codice in modo da evitare
che venga confuso l'oggetto 'screen' di Access con quello di VB, e
ritengo che occorrerà anche inserire dei riferimenti a qualche libreria
(quale?).
Aiutatemi, non mi è ancora chiaro come interagire con le API da altre
applicazioni com Access o Excel.

 ha scritto nel messaggio
news:TwRw9.106420$RO.27...@twister1.libero.it...

Se il tuo problema è quello di ridimensionare i controlli delle maschere in
funzione della risoluzione dello schermo, vai sul Sito Comune (link qui
sotto), nella Sezione Forms leggiti queste due FAQ:
1) Ridimensionamento automatico dei controlli di una maschera di Alberto
Marchi
2) Making Forms Adjust Their Size to Their Contents in Access 97 di
Charlotte Foust
Se invece vuoi cambiare la risoluzione video da codice VBA, sempre sul Sito
Comune, ma nella Sezione General leggi la FAQ di Carlo Costarella intitolata
"Modificare la risoluzione video".
--
Roberto
---------------------------------------------
il Sito Comune di it.comp.appl.access:
http://www.sitocomune.com
---------------------------------------------

 news:TwRw9.106420$
RO.27...@twister1.libero.it
:

    Ti faccio un esempio:
' ------------------------------------------------------------
' AC00
' ------------------------------------------------------------
' Descrizione:
'
'   Graphics Device Interface Utilities For Access.
'   Tests Module.
'
' ------------------------------------------------------------
' Maurizio Borrelli
' 2002-11-03
' ------------------------------------------------------------
  Option Compare Database
  Option Explicit
  Option Base 0
' ------------------------------------------------------------
  Const mc_sMyName$ = "mbMAcGDIUtils_Test"
' ------------------------------------------------------------
' ============================================================
  Public Function Test()
' ------------------------------------------------------------
  Dim oMyGDI As mbCAcGDIUtils
' ------------------------------------------------------------
  Set oMyGDI = New mbCAcGDIUtils
  With oMyGDI
    Debug.Print .TwipsPerPixelX
    Debug.Print .TwipsPerPixelY
  End With
  Set oMyGDI = Nothing
  End Function
' ============================================================
' ------------------------------------------------------------
' AC00
' ------------------------------------------------------------
' Descrizione:
'
'   Graphics Device Interface Utilities For Access.
'
' ------------------------------------------------------------
' Maurizio Borrelli
' 2002-11-03
' ------------------------------------------------------------
  Option Compare Database
  Option Explicit
  Option Base 0
' ------------------------------------------------------------
  Const mc_sMyName$ = "mbCAcGDIUtils"
' ------------------------------------------------------------
' ------------------------------------------------------------
  Const LOGPIXELSX = 88 ' Number of pixels per logical inch
                        '   along the screen width.
  Const LOGPIXELSY = 90 ' Number of pixels per logical inch
                        '   along the screen height.
' ------------------------------------------------------------
' ------------------------------------------------------------
' HDC GetDC(
'   HWND  hWnd  // handle of window
' );
' ------------------------------------------------------------
  Private Declare Function GetDC _
    Lib "user32" ( _
      ByVal hWnd As Long _
    ) As Long
' ------------------------------------------------------------
' ------------------------------------------------------------
' int GetDeviceCaps(
'   HDC hdc,    //  device-context handle
'   int nIndex  //  index of capability to query
' );
' ------------------------------------------------------------
  Private Declare Function GetDeviceCaps _
    Lib "gdi32" ( _
      ByVal hDC As Long _
    , ByVal nIndex As Long _
    ) As Long
' ------------------------------------------------------------
' ------------------------------------------------------------
' int ReleaseDC(
'   HWND  hWnd, // handle of window
'   HDC   hDc   // handle of device context
' );
' ------------------------------------------------------------
  Private Declare Function ReleaseDC _
    Lib "user32" ( _
      ByVal hWnd As Long _
    , ByVal hDC As Long _
    ) As Long
' ------------------------------------------------------------
' ------------------------------------------------------------
  Const mc_iTwipsPerInch% = 1440
' ------------------------------------------------------------
' ------------------------------------------------------------
  Private m_oAcApp          As Access.Application
  Private m_iTwipsPerPixelX As Integer
  Private m_iTwipsPerPixelY As Integer
' ------------------------------------------------------------
' ============================================================
  Public Property Get TwipsPerPixelX( _
  ) As Integer
' ------------------------------------------------------------
  TwipsPerPixelX = m_iTwipsPerPixelX
  End Property
' ============================================================
' ============================================================
  Public Property Get TwipsPerPixelY( _
  ) As Integer
' ------------------------------------------------------------
  TwipsPerPixelY = m_iTwipsPerPixelY
  End Property
' ============================================================
' ============================================================
  Private Sub Class_Initialize()
' ------------------------------------------------------------
  On Error GoTo erh
' ------------------------------------------------------------
  Dim hWnd  As Long
  Dim hDC   As Long
' ------------------------------------------------------------
  Set m_oAcApp = Access.Application
  hWnd = m_oAcApp.hWndAccessApp
  hDC = GetDC(hWnd)
  m_iTwipsPerPixelX _
    = (1 / GetDeviceCaps(hDC, LOGPIXELSX)) * mc_iTwipsPerInch
  m_iTwipsPerPixelY _
    = (1 / GetDeviceCaps(hDC, LOGPIXELSY)) * mc_iTwipsPerInch
  ReleaseDC hWnd, hDC
ext:  Exit Sub
erh:  Resume ext
  End Sub
' ============================================================
  Private Sub Class_Terminate()
' ------------------------------------------------------------
  Set m_oAcApp = Nothing
  End Sub
' ============================================================
--
Ciao :-)
Maurizio
--------
?SPQR(C)
 X
--------